Actividad 8: Iniciándose en cómputo simbólico con Maxima

Ana Gabriela Carretas Talamante

15 de abril de 2016

1. Introducción

En esta actividad se nos presentó una herramienta de cálculo bastante versátil:Maxima. Es un sistema para la manipulación de expresiones simbólicas y numéricas, incluyendo diferenciación, integración, expansión en series de Taylor, transformadas de Laplace, ecuaciones diferenciales ordinarias, sistemas de ecuaciones lineales, vectores, matrices y tensores. Maxima produce resultados de alta precisión usando fracciones exactas, números enteros de precisión arbitraria y números de coma flotante con precisión variable. Adicionalmente puede graficar funciones y datos en dos y tres dimensiones [1].



Figura 1: Interfaz de trabajo en Maxima

Se nos pidió familiarizarnos con esta herramienta de cálculo al basarnos en el manual de Jay Kerns sobre cálculo en varias variables utilizando Maxima [2]. En esta ocasión trabajé en wxMaxima, un programa que permite visualizar gráficamente los comandos utilizados en Maxima para mayor comodidad. De ahí exporté los códigos en formato `.tex` para incluirlos en la bitácora presentada a continuación.

2. Bitácora de trabajo

2.1. Geometría en tres dimensiones

Vectores y álgebra lineal En esta sección aprendimos a declarar vectores y algunas de las operaciones básicas entre ellos, como el producto punto y el producto cruz.

```
(%i1) a: [4,3,-5];
      b: [0,-1,7];
      c: [-2,-6,-1];
      d: (a.b)/(a.c);
      load(vect);
      b.(a~c);
      express(\%);
```

(%o1) $[4, 3, -5]$
(%o2) $[0, -1, 7]$
(%o3) $[-2, -6, -1]$
(%o4) $\frac{38}{21}$
(%o6) $-[0, -1, 7].[-2, -6, -1]\ [4, 3, -5]$
(%o7) -140

Líneas, planos y superficies cuadráticas Definimos ecuaciones de planos y superficies, para poder visualizarlas en una imagen.

```
(%i19) load(draw);
       cosa: x^2+y^2-z^2=3;
       draw3d(enhanced3d=true,
       implicit(cosa,x,-2,2,y,-2,2,z,-2,2),
       palette=gray);
```

(%o20) $-z^2 + y^2 + x^2 = 3$ (%o21) $[\mathrm{gr3d}\,(\mathit{implicit})]$



Figura 2: Superficie descrita por la ecuación: $-z^2 + y^2 + x^2 = 3$

Funciones vectoriales También puede calcular funciones vectoriales, parametrizarlas y graficarlas.

```
(%i1)  r(t):=[cos(t),sin(t),t];
```

(%o1) $\mathrm{r}(t) := [\cos(t), \sin(t), t]$

```
(%i2)  r(10);
```

(%o2) $[\cos(10), \sin(10), 10]$

```
(%i3)  float(%);
```

(%o3) $[-0.8390715290764524, -0.5440211108893698, 10.0]$

```
(%i4)  load(draw);
(%i23) draw3d(parametric(t,cos(t),sin(t),t,-10,10));
```

(%o23) $[\mathrm{gr3d}(parametric)]$



Figura 3: Trayectoria descrita por la ec. paramétrica: $(t, cos(t), sin(t), t) \quad t \in [-10, 10]$

```
(%i24) limit(r(t),t,2,plus);
```

(%o24) $[\cos(2), \sin(2), 2]$

```
(%i25) float(%);
```

(%o25) $[-0.4161468365471424, 0.9092974268256817, 2.0]$

```
(%i26) diff(r(t),t);
```

(%o26) $[-\sin(t), \cos(t), 1]$

```
(%i28) define(rp(t),diff(r(t),t));
```

(%o28) $\mathrm{rp}(t) := [-\sin(t), \cos(t), 1]$

```
(%i29) float(rp(10));
```

(%o29) $[0.5440211108893698, -0.8390715290764524, 1.0]$

```
(%i30) load(eigen);
(%i31) uvect(rp(t));
```

(%o31) $$[-\frac{\sin(t)}{\sqrt{\sin(t)^2+\cos(t)^2+1}}, \frac{\cos(t)}{\sqrt{\sin(t)^2+\cos(t)^2+1}}, \frac{1}{\sqrt{\sin(t)^2+\cos(t)^2+1}}]$$

```
(%i32) trigsimp(%);
```

(%o32) $$[-\frac{\sin(t)}{\sqrt{2}}, \frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$$

```
(%i33) define(T(t),%);
```

(%o33) $$\mathrm{T}(t) := [-\frac{\sin(t)}{\sqrt{2}}, \frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$$

```
(%i34) define(Tp(t),diff(T(t),t));
```

(%o34) $$\mathrm{Tp}(t) := [-\frac{\cos(t)}{\sqrt{2}}, -\frac{\sin(t)}{\sqrt{2}}, 0]$$

```
(%i35) uvect(Tp(t));
```

(%o35) $$[-\frac{\cos(t)}{\sqrt{\sin(t)^2+\cos(t)^2}}, -\frac{\sin(t)}{\sqrt{\sin(t)^2+\cos(t)^2}}, 0]$$

```
(%i36) trigsimp(%);
```

(%o36) $[-\cos(t), -\sin(t), 0]$

```
(%i37) define(N(t),%);
```

(%o37) $\mathrm{N}(t) := [-\cos(t), -\sin(t), 0]$

```
(%i38) load(vect);
(%i39) express(T(t)~N(t));
```

(%o39) $$[\frac{\sin(t)}{\sqrt{2}}, -\frac{\cos(t)}{\sqrt{2}}, \frac{\sin(t)^2}{\sqrt{2}} + \frac{\cos(t)^2}{\sqrt{2}}]$$

```
(%i40) trigsimp(%);
```

(%o40) $$[\frac{\sin(t)}{\sqrt{2}}, -\frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$$

(%i41) `define(B(t),%);`

(%o41) $B(t) := [\frac{\sin(t)}{\sqrt{2}}, -\frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$

(%i42) `float(B(10));`

(%o42) $[-0.384681016618512, 0.5933131681105248, 0.7071067811865475]$

Longitud de arco y curvatura Calculando estas cualidades de cierta ecuación paramétrica con las herramientas que nos brinda Maxima.

(%i1) `r(t):=[cos(t),sin(t),t];`

(%o1) $r(t) := [\cos(t), \sin(t), t]$

(%i2) `define(rp(t),diff(r(t),t));`

(%o2) $rp(t) := [-\sin(t), \cos(t), 1]$

(%i3) `load(eigen);`

(%i4) `uvect(rp(t));`

(%o4) $[-\frac{\sin(t)}{\sqrt{\sin(t)^2+\cos(t)^2+1}}, \frac{\cos(t)}{\sqrt{\sin(t)^2+\cos(t)^2+1}}, \frac{1}{\sqrt{\sin(t)^2+\cos(t)^2+1}}]$

(%i5) `trigsimp(%);`

(%o5) $[-\frac{\sin(t)}{\sqrt{2}}, \frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$

(%i6) `define(T(t),%);`

(%o6) $T(t) := [-\frac{\sin(t)}{\sqrt{2}}, \frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$

(%i7) `define(Tp(t),diff(T(t),t));`

(%o7) $Tp(t) := [-\frac{\cos(t)}{\sqrt{2}}, -\frac{\sin(t)}{\sqrt{2}}, 0]$

(%i8) `sqrt(Tp(t).Tp(t))/sqrt(rp(t).rp(t));`

(%o8) $\frac{\sqrt{\frac{\sin(t)^2}{2}+\frac{\cos(t)^2}{2}}}{\sqrt{\sin(t)^2+\cos(t)^2+1}}$

(%i9) `trigsimp(%);`

(%o9) $\frac{1}{2}$

```
(%i10) define(kappa(t),%);
```

(%o10) $\kappa(t) := \frac{1}{2}$

```
(%i12) integrate(r(t),t);
```

(%o12) $[\sin(t), -\cos(t), \frac{t^2}{2}]$

```
(%i13) g(t):=[4*t,-3*(t+4)^2,cos(t)];
```

(%o13) $\mathrm{g}(t) := [4\,t, (-3)\ (t+4)^2, \cos(t)]$

```
(%i14) define(gp(t),diff(g(t),t));
```

(%o14) $\mathrm{gp}(t) := [4, -6\ (t+4), -\sin(t)]$

```
(%i15) integrate(trigsimp(sqrt(gp(t).gp(t))),t,0,2*%pi);
```

(%o15) $\int_0^{2\,\pi} \sqrt{\sin(t)^2 + 36\,t^2 + 288\,t + 592}dt$

```
(%i16) romberg(sqrt(gp(t).gp(t)),t,0,2*%pi);
```

(%o16) 270.5253763666485

2.2. Funciones de varias variables

Podemos graficar también funciones de varias variables, sus curvas de nivel.

```
(%i1)  f(x,y):=(2*y*sin(2*x));
```

(%o1) $f(x,y) := 2\,y\sin(2\,x)$

```
(%i2)  load(draw);
```

```
(%i3)  draw3d(explicit(f(x,y),x,-5,5,y,-5,5));
```

(%o3) $[\mathrm{gr3d}(explicit)]$



Figura 4: Superficie descrita por la ecuación: $2\,y\sin(2\,x)$

```
(%i6)  draw3d(explicit(f(x,y),x,-5,5,y,-5,5),
       contour_levels=15,
       contour=map);
```

(%o6) $[\mathrm{gr3d}(explicit)]$



Figura 5: Curvas de nivel de la ecuación: $2\,y\sin(2\,x)$

```
(%i7) draw3d(enhanced3d=true,
      explicit(f(x,y),x,-5,5,y,-5,5),
      contour_levels=15,
      contour=surface,
      surface_hide=true);
```

(%o7) $[\mathrm{gr3d}\,(explicit)]$



Figura 6: Superficie descrita por la ecuación: $2\,y\,\sin\left(2\,x\right)$

Derivadas parciales Aprendiendo a hacer derivadas parciales con Maxima.

```
(%i1) diff(f(x,y),x);
```

(%o1) $\frac{d}{d\,x}\,\mathrm{f}\left(x,y\right)$

```
(%i3) diff(diff(f(x,y),x),x);
```

(%o3) $\frac{d^2}{d\,x^2}\,\mathrm{f}\left(x,y\right)$

```
(%i4) diff(diff(f(x,y),x),y);
```

(%o4) $\frac{d^2}{d\,x\,d\,y}\,\mathrm{f}\left(x,y\right)$

```
(%i5) A:3*x^11+y^3*x^5;
```

(%o5) $x^5\,y^3+3\,x^{11}$

```
(%i7) diff(A,x,2,y,2,x,1);
```

(%o7) $360\,x^2\,y$

Aproximación lineal y diferenciales Con la función de taylor podemos encontrar la aproximación lineal en cierto punto al orden que decidamos, y obtener diferenciales sin especificar ninguna variable independiente.

```
(%i1) f(x,y):=3*ln(x)*5*cos(y);
```

(%o1) $f(x,y) := 3\ln(x)\ 5\cos(y)$

```
(%i4) taylor(f(x,y),[x,y],[1,2],1);
```

(%o4)/T/ $15\ln(1)\cos(2) + \left(15\left(\left.\left.\frac{d}{dx}\ln(x)\right|_{x=1}\right|_{y=2}\right)\cos(2)\ (x-1) - 15\ln(1)\sin(2)\ (y-2)\right) + \ldots$

Regla de la cadena y derivación implícita Podemos hacer regla de la cadena y derivación implícita con la función `diff()`.

```
(%i1) f(x,y):=4*sin(x^3)/sin(x^2);
```

(%o1) $f(x,y) := \frac{4\sin(x^3)}{\sin(x^2)}$

```
(%i2) [x,y]:[s*t,3*t*s^3];
```

(%o2) $[s\,t, 3\,s^3\,t]$

```
(%i3) diff(f(x,y),s);
```

(%o3) $\frac{12\,s^2\,t^3\cos(s^3\,t^3)}{\sin(s^2\,t^2)} - \frac{8\,s\,t^2\cos(s^2\,t^2)\ \sin(s^3\,t^3)}{\sin(s^2\,t^2)^2}$

```
(%i4) diff(f(x,y),t);
```

(%o4) $\frac{12\,s^3\,t^2\cos(s^3\,t^3)}{\sin(s^2\,t^2)} - \frac{8\,s^2\,t\cos(s^2\,t^2)\ \sin(s^3\,t^3)}{\sin(s^2\,t^2)^2}$

```
(%i5) kill(x,y);
```

(%o5) *done*

```
(%i6) diff(f(x,y),y);
```

(%o6) 0

```
(%i7) diff(f(x,y),x);
```

(%o7) $\frac{12\,x^2\cos(x^3)}{\sin(x^2)} - \frac{8\,x\cos(x^2)\ \sin(x^3)}{\sin(x^2)^2}$

```
(%i8) F:x*y*3*z-y^4*z^4;
```

(%o8) $3\,x\,y\,z - y^4\,z^4$

```
(%i9) Fx:diff(F,x);
```

$$(\%o9)\quad 3\,y\,z$$

```
(%i10) Fy:diff(F,y);
```

$$(\%o10)\quad 3\,x\,z-4\,y^3\,z^4$$

```
(%i11) Fz:diff(F,z);
```

$$(\%o11)\quad 3\,x\,y-4\,y^4\,z^3$$

```
(%i12) [-Fx/Fz,Fy/Fz];
```

$$(\%o12)\quad [-\frac{3\,y\,z}{3\,x\,y-4\,y^4\,z^3},\frac{3\,x\,z-4\,y^3\,z^4}{3\,x\,y-4\,y^4\,z^3}]$$

Derivadas direccionales y el gradiente La función `ev()` reconoce a la expresión de gradiente en Maxima, facilitando su cálculo.

```
(%i1) f(x,y):=4*sin(x^3)/sin(x^2);
```

$$(\%o1)\quad \mathrm{f}\left(x,y\right):=\frac{4\sin\left(x^3\right)}{\sin\left(x^2\right)}$$

```
(%i2) load(vect);
```

```
(%i3) scalefactors([x,y]);
```

$$(\%o3)\quad \mathit{done}$$

```
(%i4) gdf:grad(f(x,y));
```

$$(\%o4)\quad 4\,\mathrm{grad}\left(\frac{\sin\left(x^3\right)}{\sin\left(x^2\right)}\right)$$

```
(%i5) ev(express(gdf),diff);
```

$$(\%o5)\quad [4\left(\frac{3\,x^2\cos\left(x^3\right)}{\sin\left(x^2\right)}-\frac{2\,x\cos\left(x^2\right)\,\sin\left(x^3\right)}{\sin\left(x^2\right)^2}\right),0]$$

```
(%i6) define(gdf(x,y),%);
```

$$(\%o6)\quad \mathrm{gdf}\left(x,y\right):=[4\left(\frac{3\,x^2\cos\left(x^3\right)}{\sin\left(x^2\right)}-\frac{2\,x\cos\left(x^2\right)\,\sin\left(x^3\right)}{\sin\left(x^2\right)^2}\right),0]$$

```
(%i7) v:[5,5];
```

$$(\%o7)\quad [5,5]$$

```
(%i8) (gdf(2,-5).v)/sqrt(v.v);
```

(%o8) $2^{\frac{3}{2}}\left(\frac{12\cos(8)}{\sin(4)}-\frac{4\cos(4)\sin(8)}{\sin(4)^2}\right)$

```
(%i9)  ev(%,diff);
```

(%o9) $2^{\frac{3}{2}}\left(\frac{12\cos(8)}{\sin(4)}-\frac{4\cos(4)\sin(8)}{\sin(4)^2}\right)$

```
(%i10) float(%);
```

(%o10) 19.29961717587629

```
(%i11) sqrt(gdf(1,2).gdf(1,2));
```

(%o11) $\frac{4\cos(1)}{\sin(1)}$

```
(%i12) float(ev(%,diff));
```

(%o12) 2.568370463737323

Optimización y extremos locales Podemos visualizar las curvas de nivel en una superficie o hacer el proceso con cálculo para encontrar puntos de optimización.

```
(%i1)  f(x,y):=(exp(-(x^2+y^2)));
```

(%o1) $\mathrm{f}(x,y):=\exp\left(-\left(x^2+y^2\right)\right)$

```
(%i2)  load(draw);
```

```
(%i8)  draw3d(enhanced3d=true,
       explicit(f(x,y),x,-5,5,y,-5,5),
       palette=gray);
```

(%o8) $[\mathrm{gr3d}\,(explicit)]$



Figura 7: Superficie descrita por la ecuación: $\exp\left(-\left(x^2+y^2\right)\right)$

(%i11) `draw3d(explicit(f(x,y),x,-5,5,y,-5,5), contour=map);`

(%o11) $[\mathrm{gr3d}\,(explicit)]$



Figura 8: Curvas de nivel de la superficie: $\exp\left(-\left(x^2+y^2\right)\right)$

(%i12) `fx:diff(f(x,y),x);`

(%o12) $-2\,x\,e^{-y^2-x^2}$

(%i13) `fy:diff(f(x,y),y);`

(%o13) $-2\,y\,e^{-y^2-x^2}$

(%i14) `solve([fx,fy],[x,y]);`

(%o14) $[[x=0,y=0]]$

(%i15) `H:hessian(f(x,y),[x,y]);`

(%o15) $\begin{pmatrix} 4\,x^2\,e^{-y^2-x^2}-2\,e^{-y^2-x^2} & 4\,x\,y\,e^{-y^2-x^2} \\ 4\,x\,y\,e^{-y^2-x^2} & 4\,y^2\,e^{-y^2-x^2}-2\,e^{-y^2-x^2} \end{pmatrix}$

(%i16) `determinant(H);`

(%o16) $\left(4\,x^2\,e^{-y^2-x^2}-2\,e^{-y^2-x^2}\right)\,\left(4\,y^2\,e^{-y^2-x^2}-2\,e^{-y^2-x^2}\right)-16\,x^2\,y^2\,e^{-2\,y^2-2\,x^2}$

(%i17) `subst([x=0,y=0],diff(fx,x));`

(%o17) -2

(%i18) `subst([x=0,y=0],determinant(H));`

(%o18) 4

```
(%i19) f(0,0);
```

(%o19) 1

Multiplicadores de Lagrange Podemos resolver ecuaciones fácilmente, un ejemplo de esto es para realizar el método de multiplicadores de Lagrange para optimizar superficies.

```
(%i1)  f(x,y):=exp(-(x^2+y^2));
```

(%o1) $\mathrm{f}(x,y) := \exp\left(-\left(x^2+y^2\right)\right)$

```
(%i18) g:x^2/9+y^2/4;
```

(%o18) $\frac{y^2}{4}+\frac{x^2}{9}$

```
(%i19) eq1:diff(f(x,y),x)=h*diff(g,x);
```

(%o19) $-2\,x\,e^{-y^2-x^2}=\frac{2\,h\,x}{9}$

```
(%i20) eq2:diff(f(x,y),y)=h*diff(g,y);
```

(%o20) $-2\,y\,e^{-y^2-x^2}=\frac{h\,y}{2}$

```
(%i21) eq3:g=1;
```

(%o21) $\frac{y^2}{4}+\frac{x^2}{9}=1$

```
(%i22) solve([eq1,eq2,eq3],[x,y,h]);
```

(%o22) $[[x=-3,y=0,h=-9\,e^{-9}],[x=3,y=0,h=-9\,e^{-9}],[x=0,y=2,h=-4\,e^{-4}],[x=0,y=-2,h=-4\,e^{-4}]]$

```
(%i23) [f(-3,0),f(3,0),f(0,2),f(0,-2)];
```

(%o23) $[e^{-9},e^{-9},e^{-4},e^{-4}]$

2.3. Integración múltiple

Integrales dobles Para calcular integrales está la función `integrate()`.

```
(%i1) f(x,y):=sqrt(exp(2*x)+exp(-2*x)+2);
```

(%o1) $\mathrm{f}(x,y) := \sqrt{\exp(2\,x) + \exp((-2)\;x) + 2}$

```
(%i2) integrate(integrate(f(x,y),y),x);
```

(%o2) $\left(e^{x} - e^{-x}\right)\; y$

```
(%i3) integrate(integrate(f(x,y),y,0,x^2),x,-1,1);
```

(%o3) $2\,e^{-1}\;\left(e^{2} - 5\right)$

```
(%i4) float(%);
```

(%o4) 1.757769245203667

Integración en coordenadas polares También se puede hacer en coordenadas polares.

```
(%i1) f(x,y):=x^2/4+y^2/9;
```

(%o1) $\mathrm{f}(x,y) := \frac{x^2}{4} + \frac{y^2}{9}$

```
(%i2) [x,y]:[r*cos(theta),r*sin(theta)];
```

(%o2) $[r\cos(\theta)\,, r\sin(\theta)]$

```
(%i3) integrate(integrate(f(x,y)*r,r,0,2*cos(theta)),theta,-%pi/4,%pi/2);
```

(%o3) $\frac{441\,\pi + 412}{1728}$

```
(%i4) float(%);
```

(%o4) 1.040186551060821

Integrales triples De manera similar al comando `diff()`, el `integrate()` se puede sobreponer para hacer múltiples integrales.

```
(%i1) integrate(integrate(integrate(x*y*z^4,y,0,4*x*z),x,-z+4,z^3),z,-3,3);
```

(%o1) $\frac{356185597536}{1463}$

```
(%i2) float(%);
```

(%o2) $2.43462472683527\,10^{8}$

Integrales en coordenadas cilíndricas y esféricas Para poder integrar en estas coordenadas se asignan valores a un vector, sustituyéndolos.

```
(%i1) f(x,y,z):=x*y*z;
```

(%o1) $f(x, y, z) := x\, y\, z$

```
(%i6) [x,y,z]:[r*cos(theta),r*sin(theta),z];
```

(%o6) $[r \cos(\theta), r \sin(\theta), r]$

```
(%i8) integrate(integrate(integrate(f(x,y,z)*r,z,0,5),r,0,5),theta,0,%pi/4);
```

(%o8) $\frac{3125}{4}$

```
(%i9) float(%);
```

(%o9) 781.25

```
(%i10) kill(f,x,y,z);
```

(%o10) *done*

```
(%i11) f(x,y,z):=x*y*z;
```

(%o11) $f(x, y, z) := x\, y\, z$

```
(%i12) [x,y,z]:[rho*sin(phi)*cos(theta),rho*sin(phi)*sin(theta),rho*cos(theta)];
```

(%o12) $[\sin(\phi)\ \rho \cos(\theta), \sin(\phi)\ \rho \sin(\theta), \rho \cos(\theta)]$

```
(%i13) integrate(integrate(integrate(f(x,y,z)*rho^2*sin(phi),rho,0,2),
       theta,0,%pi/2),phi,0,%pi/4);
```

(%o13) $\frac{32 \left(\frac{2}{3} - \frac{5}{3\, 2^{\frac{3}{2}}}\right)}{9}$

```
(%i14) float(%);
```

(%o14) 0.2752391668546742

Cambio de variables Podemos aplicar el teorema de cambio de variables sin tener que hacer una cuenta manualmente.

```
(%i1) f(x,y):=3*x+y^2;
```

(%o1) $f(x, y) := 3\, x + y^2$

```
(%i2) [x,y]:[3*u+4*v,u^2+y^2];
```

(%o2) $[4\,v+3\,u,y^2+u^2]$

```
(%i3) J:jacobian([x,y],[u,v]);
```

(%o3) $\begin{pmatrix} 3 & 4 \\ 2\,u & 0 \end{pmatrix}$

```
(%i4) J:determinant(J);
```

(%o4) $-8\,u$

```
(%i5) J:J*(-1);
```

(%o5) $8\,u$

```
(%i6) integrate(integrate(f(x,y)*J,u,0,2),v,1,2);
```

(%o6) $\dfrac{8\,(6\,y^4+24\,y^2+212)}{3}$

2.4. Cálculo vectorial

Campos vectoriales dos-dimensionales Este campo lo graficamos en el paquete `draw` de Maxima.

```
(%i2) coord: setify(makelist(k,k,-5,5));
      points2d: listify(cartesian_product(coord,coord));
      vf2d(x,y):=vector([x,y],[x+y,y^2]/10);
      vect2: makelist(vf2d(k[1],k[2]),k,points2d);
      load(draw);
      apply(draw2d,append([head_length=0.1,color=red],vect2));
```



Figura 9: Campo vectorial de la función $f(x,y)=(x+y,y^2)$

Campos vectoriales de gradiente Retomando la manera en que calculamos el gradiente de una función, podemos obtener el campo que describe.

(%i1) `load(vect);`

(%i2) `f(x,y):=x^2/4+y^2/9;`

(%o2) $\mathrm{f}(x,y) := \frac{x^2}{4} + \frac{y^2}{9}$

(%i3) `scalefactors([x,y]);`

(%o3) *done*

(%i4) `gdf(x,y):=grad(f(x,y));`

(%o4) $\mathrm{gdf}(x,y) := \mathrm{grad}(\mathrm{f}(x,y))$

(%i5) `ev(express(gdf(x,y)),diff);`

(%o5) $[\frac{x}{2}, \frac{2\,y}{9}]$

(%i6) `define(gdf(x,y),%);`

(%o6) $\mathrm{gdf}(x,y) := [\frac{x}{2}, \frac{2\,y}{9}]$

(%i7) `load(draw);`

(%i8)
```
coord: setify(makelist(k,k,-5,5));
points2d: listify(cartesian_product(coord,coord));
vf2d(x,y):=vector([x,y],gdf(x,y)/5);
vect2: makelist(vf2d(k[1],k[2]),k,points2d);
apply(draw2d,append([head_length=0.1,color=red],vect2));
```



Figura 10: Campo vectorial del gradiente de la función $f(x,y) = (\frac{x^2}{4} + \frac{y^2}{9})$

Campos vectoriales tres-dimensionales Y también pueden no ser en dos dimensiones.

```
(%i1)  load(vect);

(%i2)  load(draw);

(%i5)  coord: setify(makelist(k,k,-5,5));
       points3d: listify(cartesian_product(coord,coord,coord));
       vf3d(x,y,z):=vector([x,y,z],[2*x,2*y,2*z]/5);
       vect3: makelist(vf3d(k[1],k[2],k[3]),k,points3d);
       apply(draw3d,append([head_length=0.1,color=red],vect3));
```



Figura 11: Campo vectorial de la función $f(x, y) = (2x, 2y, 2z)$

Integrales de línea respecto a la longitud de arco Con la función `romberg()` podemos calcular integrales de línea al tener una función parametrizada.

```
(%i1)  f(x,y):=x^2/4+y^2/9;
```

(%o1) $\mathrm{f}(x, y) := \frac{x^2}{4} + \frac{y^2}{9}$

```
(%i2)  [x,y]:[2*cos(t),3*sin(t)];
```

(%o2) $[2\cos(t), 3\sin(t)]$

```
(%i3)  rp:diff([x,y],t);
```

(%o3) $[-2\sin(t), 3\cos(t)]$

```
(%i4)  romberg(f(x,y)*sqrt(rp.rp),t,1,2);
```

(%o4) 2.099777455250255

Integrales de línea de campos vectoriales De la misma forma, la función `romberg()` admite calcular integrales con vectores.

```
(%i1)  F(x,y,z):=[x*y*z,3*x*y^2,z];
```

(%o1) $F(x, y, z) := [x\,y\,z, 3\,x\,y^2, z]$

```
(%i2)  [x,y,z]:[t+2,t/%pi,t^5];
```

(%o2) $[t+2, \frac{t}{\pi}, t^5]$

```
(%i3)  romberg(F(x,y,z).diff([x,y,z],t),t,0,%pi);
```

(%o3) 47480.59921273958

Campos vectoriales conservativos y encontrando potenciales escalares Con la función `curl()` podemos notar si los campos son o no conservativos, y si lo son, podemos encontrar el potencial escalar con la función `potential()`. Ambas se encuentran en el paquete `vect`, que hemos estado utilizando a lo largo de esta actividad.

```
(%i1)  load(vect);
(%i2)  F(x,y):=[x^2+y^2,2*y+3*x];
```

(%o2) $F(x, y) := [x^2 + y^2, 2\,y + 3\,x]$

```
(%i3)  scalefactors([x,y]);
```

(%o3) *done*

```
(%i4)  curl(F(x,y));
```

(%o4) $\operatorname{curl}\left([y^2 + x^2, 2\,y + 3\,x]\right)$

```
(%i5)  express(%);
```

(%o5) $\frac{d}{d\,x}\,(2\,y + 3\,x) - \frac{d}{d\,y}\,\left(y^2 + x^2\right)$

```
(%i6)  ev(%,diff);
```

(%o6) $3 - 2\,y$

```
(%i15) F(x,y,z):=[y,z*cos(y*z)+x,y*cos(y*z)];
```

(%o15) $F(x, y, z) := [y, z\cos(y\,z) + x, y\cos(y\,z)]$

```
(%i17) ev(express(curl(F(x,y,z))),diff);
```

(%o17) $[0, 0, 0]$

```
(%i14) F(u,v,w):=[v,w*cos(v*w)+u,v*cos(v*w)];
```

(%o14) $\mathrm{F}(u,v,w) := [v, w\cos(v\,w)+u, v\cos(v\,w)]$

```
(%i8)  scalefactors([u,v,w]);
```

(%o8) *done*

```
(%i9)  potential(F(u,v,w));
```

(%o9) $\sin(v\,w)+u\,v$

```
(%i10) define(f(u,v,w),%);
```

(%o10) $\mathrm{f}(u,v,w) := \sin(v\,w)+u\,v$

```
(%i12) f(5,1,2)-f(1,1,1);
```

(%o12) $\sin(2)-\sin(1)+4$

```
(%i13) float(%);
```

(%o13) 4.067826442017785

Referencias

[1] Source Forge, *Maxima*. Recuperado en abril de 2016 de http://maxima.sourceforge.net/es/

[2] Kerns, Jay. *Multivariable Calculus with Maxima*. Recuperado en abril de 2016 de http://gkerns.people.ysu.edu/maxima/maximaintro/

[3] Lizárraga, C. *Actividad 8 (2016-1)*. Recuperado en abril de 2016 de http://computacional1.pbworks.com/w/page/106192917/Actividad%208%20(2016-1)